/*
 * Licensed under the MIT license
 *
 * Copyright (c) 2009-2011 Pieter van de Sluis
 *
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 *
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 *
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 *
 * http://code.google.com/p/imotionproductions/
 */

package nl.imotion.notes
{
    import flash.utils.Dictionary;


    public class NoteDispatcher
	{
		private static var allowInstantiation:Boolean = false;
		private static var instance:NoteDispatcher;
		
		private var listenerMap:Dictionary = new Dictionary();
		
		
		public function NoteDispatcher()
		{
			if ( !allowInstantiation )
			{
				throw new Error( "Instantiation failed: Use NoteDispatcher.getInstance() instead of constructor." );
			}
		}
		
		
		public static function getInstance():NoteDispatcher
		{
			if ( instance == null )
			{
				allowInstantiation = true;
				instance = new NoteDispatcher();
				allowInstantiation = false;
			}
			return instance as NoteDispatcher;
		}
		
		
		public function addNoteListener( type:String, listener:Function ):void
		{
			var typeListeners:Array = listenerMap[ type ];
			
			if ( typeListeners ) 
			{
				for each( var f:Function in typeListeners )
				{
					if ( listener == f )
					{
						return;
					}
				}
				typeListeners.push( listener );
			} 
			else 
			{
				listenerMap[ type ] = [ listener ];	
			}
		}
		
		
		public function removeNoteListener( type:String, listener:Function ):void
		{
			var typeListeners:Array = listenerMap[ type ];
			
			if ( typeListeners )
			{
				for ( var i:int = 0; i < typeListeners.length; i++ ) 
				{
					if ( typeListeners[ i ] == listener )
					{
						typeListeners.splice( i, 1 );
						break;
					}
				}
				
				if ( typeListeners.length == 0)
				{
					delete listenerMap[ type ];
				}
			}
		}
		
		
		public function hasNoteListener( type:String ):Boolean
		{
			return ( listenerMap[ type ] != null );
		}
		
		
		public function dispatchNote( note:Note ):void
		{
			var typeListeners:Array = listenerMap[ note.type ];
			
			for each( var f:Function in typeListeners )
			{
				f( note );
			}
		}
		
	}
	
}